



Analytic Guidelines: Creating Disability Identifiers Using the Washington Group Extended Set (WG-ES) SPSS Syntax

Introduction

As with the WG Short Set (WG-SS), analysis of the WG Extended Set (WG-ES) can also produce multiple disability identifiers based on the choice of the severity threshold or cut-off. The SPSS syntax below, however, provides for the calculation of disability identifiers using different sets of WG-ES domains using the recommended cut-off for international comparisons (described below).

For each of the disability identifiers described, the level of inclusion is at least one domain/question is coded A LOT OF DIFFICULTY or CANNOT DO AT ALL – or – for the domains Anxiety, Depression, Pain and Fatigue, the highest level of difficulty on a four-point scale.

Each of the four disability identifiers described in this document is defined based on the choice of domains of functioning included:

WG-SS: *Short Set:* 6 domains, 6 questions.

WG-ES 1: *Extended Set:* 11 domains, 25 questions.

WG-ES 2: *Modified Extended Set (WG-ES MINUS Pain and Fatigue):* 9 domains, 20 questions.

WG-ES 3: *Short Set Enhanced (WG-SS PLUS Upper body, Anxiety and Depression):* 9 domains, 12 questions.

NOTE: For data analysis, use your standard weighting and estimation techniques.

The SPSS syntax is based on the *variable labels* indicated in the table below. The complete WG-ES module includes more questions than appear in this table. Disability status is determined through difficulty in the basic, universal activities *without* the use of assistive technology or other assistance. There are several mobility questions, for example, that reference difficulty walking *with* the use of assistance. Those questions are not included in the analytic plan provided here; however, they can be

This sixth in a series of **Washington Group Implementation Documents** covers analytic guidance, including SPSS syntax, when using the Washington Group Extended Set questions. Guidance on creating disability identifiers is provided.

Additional Implementation Documents cover: The Washington Group Tools; The Washington Group Short Set on Functioning; Translation of the WG Tools; Question Specifications; and other tools developed by the WG and partners for the measurement of disability.

For more information, visit the Washington Group website: <http://www.washingtongroup-disability.com/>.

Analytic Guidelines: The Washington Group Extended Set on Functioning

used in other analyses that look more closely into the effect of assistive technology (environmental facilitators) on functioning.

Only those questions/variables below are used in the determination of disability identifiers. Ensure that you use the same *variable labels* OR revise the SPSS syntax to reflect the *variable labels* in your database.

The WG-ES Questions as they appear in the U.S. National Health Interview Survey (NHIS):

WG Extended Set Questions/Domains	Variable Label	Response Pattern
VISION		
1. Do you have difficulty seeing even if wearing glasses?	VIS_SS	1
COMMUNICATION		
2. Using your usual language, do you have difficulty communicating (for example understanding or being understood by others)?	COM_SS	1
HEARING		
3. Do you have difficulty hearing even if using a hearing aid?	HEAR_SS	1
4. Difficulty hearing conversation with one person in quiet room?	HEAR_3	1
5. Difficulty hearing conversation with one person in noisier room?	HEAR_4	1
COGNITION		
6. Do you have difficulty remembering or concentrating?	COG_SS	1
7. Difficulty remembering, concentrating, or both?	COG_1	2
8. How often have difficulty remembering?	COG_2	3
9. Amount of things you have difficulty remembering?	COG_3	4
SELF-CARE / UPPER BODY		
10. Do you have difficulty with (self-care such as) washing all over or dressing?	UB_SS	1
11. Difficulty raising 2 liter bottle of water from waist to eye level?	UB_1	1
12. Degree of difficulty using hands and fingers	UB_2	1
MOBILITY		

13. Do you have difficulty walking or climbing stairs?	MOB_SS	1
14. Difficulty walking 100 yards on level ground without aid or equipment?	MOB_4	1
15. Difficulty walking 1/3rd mile on level ground without aid or equipment	MOB_5	1
16. Difficulty walking up or down 12 steps without aid or equipment?	MOB_6	1
AFFECT (ANXIETY)		
17. How often feel worried, nervous, or anxious?	ANX_1	5
18. Level of feelings when last felt worried, nervous, or anxious?	ANX_3	6
AFFECT (DEPRESSION)		
19. How often do you feel depressed?	DEP_1	5
20. How depressed you felt last time you were depressed?	DEP_3	6
PAIN		
21. Frequency of pain in past 3 months?	PAIN_2	7
22. How much pain you had last time you had pain?	PAIN_4	6
FATIGUE		
23. How often felt very tired or exhausted in past 3 months?	TIRED_1	7
24. How long most recent tired or exhausted feelings lasted?	TIRED_2	8
25. Level of tiredness last time felt very tired or exhausted?	TIRED_3	6

NOTE: **Red** refers to the Washington Group Short Set (**WG-SS**).
All 25 questions are included in **WG-ES 1**.

Red plus **Blue** plus **Green** questions are included in **WG-ES 2**.

Red plus **Green** questions are included in **WG-ES 3**.

Response patterns:

	Pattern 1	Pattern 2	Pattern 3	Pattern 4
1	No difficulty	Difficulty remembering only	Sometimes	A few things
2	Yes, Some difficulty	Difficulty concentrating only	Often	A lot of things
3	Yes, A lot of difficulty	Difficulty both remembering & concentrating	All of the time	Almost everything
4	Cannot do at all			
7	Refused			
8	Not ascertained			
9	Don't know			

	Pattern 5	Pattern 6*	Pattern 7	Pattern 8
1	Daily	A little	Never	Some of the day
2	Weekly	A lot	Some days	Most of the day
3	Monthly	Somewhere in between a little and a lot	Most days	All of the day
4	A few times a year		Every day	
5	Never			
7	Refused			
8	Not ascertained			
9	Don't know			

*** IN THE SYNTAX BELOW, NOTE THAT ITEMS WITH RESPONSE PATTERN 6 (ANX_3, DEP_3, PAIN_4 AND TIRED_3) ARE RECODED TO PLACE “SOMEWHERE BETWEEN” NUMERICALLY IN-BETWEEN “A LITTLE” AND “A LOT”.**

The WG-SS is embedded within the WG-ES.

The WG-ES is supplemented with:

- additional questions to those existing 6 domains and
- additional domains (several with multiple questions).

The SPSS syntax presented below includes a couple of elements that were particular to the content of the WG-ES.

First, it was important to determine single domain-specific identifiers for those domains of functioning that included multiple questions. For example, upper body functioning includes two questions, each eliciting specific and unique actions: difficulty raising a bottle of water from waist to eye level (arms/shoulders), and difficulty using hands and fingers. Those two questions were analyzed and combined to produce a single upper body indicator with four levels of difficulty ranging from 1 - low difficulty to 4 - high difficulty – not unlike the categorical responses to the single WG-SS questions: no difficulty, some difficulty, a lot of difficulty and cannot do at all. As with the upper body domain, the WG-ES domains cognition, anxiety, depression, pain and fatigue have different response patterns that do not readily ‘translate’ into the usual WG response pattern. For these domains of functioning, a similar 4-scale response pattern was produced and annotated as level 1 through 4, where 1 is the lowest level of difficulty and 4 is the highest.

Second, individual domain indicators were assessed to determine the appropriate cut-off for inclusion into an overall disability identifier – for the purposes of estimating prevalence and disaggregating outcome indicators by disability status.

SPSS WG Extended Set Syntax Annotated with Output Tables

Actual SPSS syntax is indented – commands in **bold text**.

NOTE: For data analysis, use your standard weighting and estimation techniques.

Step 1. Generate frequency distribution for each domain question: Vision

The syntax below produces **frequency distributions** on individual domain questions – and calculates INDICATOR variables for domains with multiple questions.

VISION

No recoding – the Short Set question is used in all calculations

```
FREQUENCIES VIS_SS
```

		VIS_SS			
		Frequency	Percent	Valid Percent	Cumulative Percent
Valid	No difficulty	13690	79.0	81.6	81.6
	Some difficulty	2708	15.6	16.2	97.8
	A lot of difficulty	333	1.9	2.0	99.8
	Cannot do at all	36	.2	.2	100.0
	Total	16767	96.8	100.0	
Missing	Refused	21	.1		
	Not ascertained	532	3.1		
	Don't know	6	.0		
	Total	559	3.2		
Total		17326	100.0		

Step 2. Generate frequency distribution for each domain question: Communication

COMMUNICATION

No recoding – the Short Set question is used in all calculations

FREQUENCIES COM_SS

		COM_SS			
		Frequency	Percent	Valid Percent	Cumulative Percent
Valid	No difficulty	15874	91.6	94.7	94.7
	Some difficulty	745	4.3	4.4	99.2
	A lot of difficulty	94	.5	.6	99.7
	Cannot do at all	43	.2	.3	100.0
	Total	16756	96.7	100.0	
Missing	Refused	25	.1		
	Not ascertained	543	3.1		
	Don't know	2	.0		
	Total	570	3.3		
Total		17326	100.0		

Step 3. Generate frequency distribution for each domain question (including domains with multiple questions): Hearing

HEARING

FREQUENCIES HEAR_SS HEAR_3 HEAR_4.

		HEAR_SS		Valid Percent	Cumulative Percent
		Frequency	Percent		
Valid	No difficulty	13680	79.0	79.0	79.0
	Some difficulty	2753	15.9	15.9	94.8
	A lot of difficulty	310	1.8	1.8	96.6
	Cannot do at all	23	.1	.1	96.8
	Refused	24	.1	.1	96.9
	Not ascertained	534	3.1	3.1	100.0
	Don't know	2	.0	.0	100.0
	Total	17326	100.0	100.0	

HEAR_3: Difficulty hearing conversation with one person in quiet room

		Frequency	Percent	Valid Percent	Cumulative Percent
Valid	No difficulty	15249	88.0	90.9	90.9
	Some difficulty	1316	7.6	7.8	98.8
	A lot of difficulty	162	.9	1.0	99.7
	Cannot do at all	10	.1	.1	99.8
	Refused	27	.2	.2	100.0
	Not ascertained	1	.0	.0	100.0
	Don't know	4	.0	.0	100.0
	Total	16769	96.8	100.0	
Missing	557	3.2			
Total	17326	100.0			

HEAR_4: Diff hearing one person in noisier room even w/ hearing aid

		Frequency	Percent	Valid Percent	Cumulative Percent
Valid	No difficulty	11697	67.5	69.8	69.8
	Some difficulty	4191	24.2	25.0	94.8
	A lot of difficulty	779	4.5	4.6	99.5
	Cannot do at all	48	.3	.3	99.7
	Refused	28	.2	.2	99.9
	Don't know	15	.1	.1	100.0
	Total	16758	96.7	100.0	
Missing		568	3.3		
Total		17326	100.0		

Step 4. For Hearing questions, recode missing values (7) Refused (8) Not ascertained and (9) Don't know into a single missing value coded as (9).

The syntax below recodes missing values (7) Refused (8) Not ascertained and (9) Don't know into a single missing value coded as (9).

RECODE HEAR_SS, HEAR_3 and HEAR_4 where refused (7) non ascertained (8) and Don't know (9) are combined into (9)

RECODE HEAR_SS (7, 8, 9 = 9).
RECODE HEAR_3 (7, 8, 9 = 9).
RECODE HEAR_4 (7, 8, 9 = 9).

Step 5. Generate a cross-tabulation of the two Hearing Extended Set questions: HEAR_3 and HEAR_4.

The syntax below produces a cross-tabulation of the two Extended Set questions: HEAR_3: Difficulty hearing in a quiet room and HEAR_4: Difficulty hearing in a noisier room to determine a single HEARING INDICATOR.

CROSSTABS HEAR_4 BY HEAR_3.

HEAR_4 (Difficulty hearing in a Nosier room)		HEAR_3 (Difficulty hearing in a Quiet room)				Missing	Total
		No difficulty	Some difficulty	A lot of difficulty	Cannot do at all		
HEAR_4x	No difficulty	11603	94	0	0	0	11697
	Some difficulty	3373	809	8	0	1	4191
	A lot of difficulty	253	388	138	0	0	779
	Cannot do at all	8	24	16	23	0	71
	Missing	12	1	0	0	30	43
Total		15249	1316	162	23	31	16781

Step 6. Create a HEARING INDICATOR (H_INDICATOR) based on the two additional hearing questions HEAR_3 and HEAR_4.

The syntax below creates a HEARING INDICATOR (H_INDICATOR) based on the two additional hearing questions HEAR_3 and HEAR_4.

```

COMPUTE H_INDICATOR = 0.
IF (HEAR_3 = 1 AND HEAR_4 = 1) OR (HEAR_3 = 1 AND HEAR_4 = 2) H_INDICATOR = 1.
IF (HEAR_3 = 2 AND (HEAR_4 = 1 OR HEAR_4 = 2)) OR (HEAR_3 = 1 AND HEAR_4 = 3) H_INDICATOR = 2.
IF (HEAR_3 = 3 AND (HEAR_4 = 1 OR HEAR_4 = 2) OR (HEAR_3 = 2 AND HEAR_4 = 3) OR (HEAR_3 = 1 AND HEAR_4 = 4)) H_INDICATOR = 3.
IF ((HEAR_3 = 3 AND HEAR_4 = 3) OR HEAR_3 = 4 OR (HEAR_4 = 4 AND (HEAR_3 = 2 OR HEAR_3 = 3))) H_INDICATOR = 4.
    
```

FREQUENCIES H_INDICATOR.

		H_INDICATOR			
		Frequency	Percent	Valid Percent	Cumulative Percent
Valid	1.00	14976	86.4	89.4	89.4
	2.00	1156	6.7	6.9	96.3
	3.00	404	2.3	2.4	98.7
	4.00	211	1.2	1.3	100.0
	Total	16747	96.7	100.0	
Missing	.00	579	3.3		
Total		17326	100.0		

Step 7. Generate frequency distribution for each domain question: Cognition

COGNITION: Degree of difficulty remembering or concentrating

Frequency distribution of the WG-SS cognition question: COG_SS.

FREQUENCIES COG_SS.

		COG_SS		Valid	Cumulative
		Frequency	Percent	Percent	Percent
Valid	No difficulty	13719	79.2	79.2	79.2
	Some difficulty	2632	15.2	15.2	94.4
	A lot of difficulty	382	2.2	2.2	96.6
	Cannot do at all	20	.1	.1	96.7
	Refused	25	.1	.1	96.8
	Not ascertained	543	3.1	3.1	100.0
	Don't know	5	.0	.0	100.0
	Total	17326	100.0	100.0	

If response to COG-SS 2, 3 or 4 then the respondent is asked COG_1: whether they have difficulty remembering, concentrating or both.

FREQUENCIES COG_1.

COG_1: Difficulty remembering, concentrating, or both?

		Frequency	Percent	Valid	Cumulative
				Percent	Percent
Valid	Difficulty remembering only	983	5.7	32.1	32.1
	Difficulty concentrating only	388	2.2	12.7	44.7
	Difficulty with both remembering and concentrating	1659	9.6	54.1	98.9
	Refused	28	.2	.9	99.8
	Don't know	6	.0	.2	100.0
	Total	3064	17.7	100.0	
Missing		14262	82.3		
Total		17326	100.0		

Step 8. Account for those who did not answer COG_1 (COG_SS is 1 – no difficulty and they were skipped) by recoding COG_1 to 0 (No difficulty).

If response to COG-SS is 1: no difficulty, then the variable COG_1 is recoded into COG_1A, and the value assigned is 0: no difficulty.

```
COMPUTE COG_1A = COG_1.
IF (COG_SS = 1) COG_1A = 0.
FREQUENCIES COG_1A.
```

		COG_1A		Valid	Cumulative
		Frequency	Percent	Percent	Percent
Valid	(0) No difficulty	13719	79.2	81.7	81.7
	Difficulty remembering only	983	5.7	5.9	87.6
	Difficulty concentrating only	388	2.2	2.3	89.9
	Difficulty with both remembering and concentrating	1659	9.6	9.9	99.8
	Refused	28	.2	.2	100.0
	Don't know	6	.0	.0	100.0
	Total	16783	96.9	100.0	
Missing		543	3.1		
Total		17326	100.0		

Step 9. Generate frequency distribution for remaining cognition questions.

Frequency distribution of the Cognition extended REMEMBERING questions: COG_2 and COG_3.

```
FREQUENCIES COG_2 COG_3.
```

COG_2: How often have difficulty remembering?

		Frequency	Percent	Valid	Cumulative
				Percent	Percent
Valid	Sometimes	1916	11.1	71.6	71.6
	Often	513	3.0	19.2	90.8
	All of the time	216	1.2	8.1	98.8
	Refused	26	.2	1.0	99.8
	Don't know	5	.0	.2	100.0
	Total	2676	15.4	100.0	
Missing		14650	84.6		
Total		17326	100.0		

COG_3: Amount of things you have difficulty remembering?

		Frequency	Percent	Valid Percent	Cumulative Percent
Valid	A few things	2119	12.2	79.2	79.2
	A lot of things	386	2.2	14.4	93.6
	Almost everything	134	.8	5.0	98.6
	Refused	26	.2	1.0	99.6
	Don't know	11	.1	.4	100.0
	Total	2676	15.4	100.0	
	Missing		14650	84.6	
Total		17326	100.0		

Step 10. Generate cross-tabulation of the two Cognition extended set questions COG_2 by COG_3.

The syntax below produces a cross-tabulation of the two Extended Set REMEMBERING questions: COG_2: How often you have difficulty remembering and COG_3: The amount of things you have difficulty remembering to determine a single REMEMBERING INDICATOR.

CROSSTABS COG_2 BY COG_3.

		COG_3: Amount of things you have difficulty remembering?					Total	
		A few things	A lot of things	Almost everything	Refused	Don't know		
COG_2: How often do you have difficulty remembering?	How often have difficulty remembering?	Sometimes	1788	105	20	1	2	1916
	Often	279	197	34	0	3	513	
	All of the time	51	84	80	0	1	216	
	Refused	1	0	0	25	0	26	
	Don't know	0	0	0	0	5	5	
Total		2119	386	134	26	11	2676	

Step 11. Create a Remembering Indicator based on distribution of COG_2 and COG_3.

The syntax below creates a REMEMBERING INDICATOR (R_INDICATOR) based on the two additional remembering questions (COG_2 and COG_3).

If COG_SS is 1: no difficulty, then the Remembering Indicator is coded as 1: the lowest level of difficulty.

COMPUTE R_INDICATOR = 0.
IF (COG_SS = 1) R_INDICATOR = 1.
IF ((COG_2 = 1 AND COG_3 = 1) OR (COG_3 = 1 AND COG_2 = 2) OR (COG_3 = 2 AND COG_2 = 1)) R_INDICATOR = 2.
IF (COG_3 = 2 AND COG_2 = 2) R_INDICATOR = 3.
IF (COG_3 = 3 OR COG_2 = 3) R_INDICATOR = 4.
FREQUENCIES R_INDICATOR.

Step 12. If COG_1A is coded as 2 (concentrating only), then the Remembering Indicator is coded as 5.

These 388 individuals are respondents who were not included in the Remembering Indicator since they had only difficulty concentrating.

IF (COG_1A = 2) R_INDICATOR = 5.

Step 13. Generate frequency distribution of the Remembering Indicator.

FREQUENCIES R_INDICATOR.

		R_INDICATOR			
		Frequency	Percent	Valid Percent	Cumulative Percent
Valid	.00	580	3.3	3.3	3.3
	1.00	13719	79.2	79.2	82.5
	2.00	2172	12.5	12.5	95.1
	3.00	197	1.1	1.1	96.2
	4.00	270	1.6	1.6	97.8
	5.00	388	2.2	2.2	100.0
	Total	17326	100.0	100.0	

Step 14. Supplement Remembering Indicator with information on difficulty concentrating.

The syntax below adds information on whether cognitive difficulties are compounded by difficulty concentrating in addition to difficulty remembering.

Create a COGNITION INDICATOR (COG_INDICATOR) based on R_INDICATOR (above) and the cognition question (COG_1).

The 388 individuals with ‘concentrating only’ were allocated as follows:

1. 357 with a little difficulty on COG_SS question were classified as **2**
2. 30 with a lot of difficulty on COG_SS question were classified as **3**
3. 1 with cannot do on COG_SS question was classified as **4**

Those with both remembering and concentrating difficulty were upgraded 36 individuals from **2** to **3**, and 125 individuals from **3** to **4**.

COMPUTE COG_INDICATOR = 0.
COMPUTE COG_INDICATOR = R_INDICATOR.
IF (R_INDICATOR = 5 AND COG_SS = 2) COG_INDICATOR = 2.
IF (R_INDICATOR = 5 AND COG_SS = 3) COG_INDICATOR = 3.
IF (R_INDICATOR = 5 AND COG_SS = 4) COG_INDICATOR = 4.

IF (R_INDICATOR = 2 AND COG_1 = 3 AND COG_SS = 3) COG_INDICATOR = 3.
IF (R_INDICATOR = 3 AND COG_1 = 3 AND COG_SS = 3) COG_INDICATOR = 4.

Step 15. Generate frequency distribution of the Cognition Indicator.

FREQUENCIES COG_INDICATOR.

		COG_INDICATOR			
		Frequency	Percent	Valid Percent	Cumulative Percent
Valid	.00	580	3.3	3.3	3.3
	1.00	13719	79.2	79.2	82.5
	2.00	2449	14.1	14.1	96.7
	3.00	226	1.3	1.3	98.0
	4.00	352	2.0	2.0	100.0
Total		17326	100.0	100.0	

UPPER BODY

Step 16. Generate frequency distribution for each domain question: Self-care and Upper body functioning.

UB_SS is the WG-SS Self-care question.

First, calculate frequency distributions on the short set and two extended set questions.

FREQUENCIES UB_SS UB_1 UB_2.

UB_SS: Degree of difficulty with self-care

		Frequency	Percent	Valid Percent	Cumulative Percent
Valid	No difficulty	16029	92.5	95.7	95.7
	Some difficulty	544	3.1	3.2	98.9
	A lot of difficulty	114	.7	.7	99.6
	Cannot do at all	68	.4	.4	100.0
	Total	16755	96.7	100.0	
Missing	Refused	25	.1		
	Not ascertained	544	3.1		
	Don't know	2	.0		
	Total	571	3.3		
Total		17326	100.0		

UB_1: Diff raising 2 liter bottle of water from waist to eye level

		Frequency	Percent	Valid Percent	Cumulative Percent
Valid	No difficulty	15677	90.5	90.5	90.5
	Some difficulty	743	4.3	4.3	94.8
	A lot of difficulty	167	1.0	1.0	95.7
	Cannot do at all	166	1.0	1.0	96.7
	Refused	24	.1	.1	96.8
	Not ascertained	545	3.1	3.1	100.0
	Don't know	4	.0	.0	100.0
	Total	17326	100.0	100.0	

UB_2: Degree of difficulty using hands and fingers

		Frequency	Percent	Valid Percent	Cumulative Percent
Valid	No difficulty	15199	87.7	87.7	87.7
	Some difficulty	1229	7.1	7.1	94.8
	A lot of difficulty	255	1.5	1.5	96.3
	Cannot do at all	70	.4	.4	96.7
	Refused	25	.1	.1	96.8
	Not ascertained	545	3.1	3.1	100.0
	Don't know	3	.0	.0	100.0
	Total	17326	100.0	100.0	

Step 17. Generate a cross-tabulation of the two Upper body Extended Set questions: UB_2 and UB_1.

The syntax below produces a cross-tabulation of the two Extended Set questions: UB_1: Difficulty raising a 2 liter bottle of water from waste to eye level UB_2: Difficulty using hands and fingers to determine a single UPPER BODY INDICATOR (UB_INDICATOR).

CROSSTABS UB_2 BY UB_1.

		UB_1: Diff raising 2 liter bottle of water from waist to eye level							Total
		No difficulty	Some difficulty	A lot of difficulty	Cannot do at all	Refused	Not ascertained	Don't know	
UB_2: Difficulty using hands and fingers	No difficulty	14786	309	58	44	1	0	1	15199
	Some difficulty	782	355	51	40	0	0	1	1229
	A lot of difficulty	98	73	51	33	0	0	0	255
	Cannot do at all	9	5	7	49	0	0	0	70
	Refused	2	0	0	0	23	0	0	25
	Not ascertained	0	0	0	0	0	545	0	545
	Don't know	0	1	0	0	0	0	2	3
	Total	15677	743	167	166	24	545	4	17326

Step 18. Create an UPPER BODY INDICATOR (UB_INDICATOR) based on the two additional hearing questions UB_2 and UB_3.

Syntax below creates UB_INDICATOR based on the distribution in the cross-tabulation above.

```

COMPUTE UB_INDICATOR = 0.
IF (UB_1 = 4 OR UB_2 = 4) UB_INDICATOR = 4.
IF UB_INDICATOR NE 4 AND (UB_1 = 3 OR UB_2 = 3) UB_INDICATOR = 3.
IF UB_INDICATOR NE 4 AND UB_INDICATOR NE 3 AND (UB_1 = 2 OR UB_2 = 2)
UB_INDICATOR = 2.
IF UB_INDICATOR NE 4 AND UB_INDICATOR NE 3 AND UB_INDICATOR NE 2 AND
(UB_1 = 1 OR UB_2 = 1) UB_INDICATOR = 1.

FREQUENCIES UB_INDICATOR.
    
```


UB_INDICATOR

		Frequency	Percent	Valid Percent	Cumulative Percent
Valid	1.00	14790	85.4	88.3	88.3
	2.00	1448	8.4	8.6	96.9
	3.00	331	1.9	2.0	98.9
	4.00	187	1.1	1.1	100.0
	Total	16756	96.7	100.0	
Missing		570	3.3		
Total		17326	100.0		

MOBILITY

Step 19. Generate frequency distribution for each domain question: Mobility.

MOB_SS is the WG-SS Mobility question.

First, calculate frequency distributions on the short set and two extended set WALKING questions.

FREQUENCIES MOB_SS MOB_4 MOB_5.

MOB_SS

		Frequency	Percent	Valid Percent	Cumulative Percent
Valid	No difficulty	13424	77.5	80.1	80.1
	Some difficulty	2165	12.5	12.9	93.0
	A lot of difficulty	792	4.6	4.7	97.7
	Cannot do at all	380	2.2	2.3	100.0
	Total	16761	96.7	100.0	
Missing	Refused	25	.1		
	Not ascertained	536	3.1		
	Don't know	4	.0		
	Total	565	3.3		
Total		17326	100.0		

MOB_4: Diff walking 100 yards on level ground w/o aid or equipment

		Frequency	Percent	Valid Percent	Cumulative Percent
Valid	No difficulty	13892	80.2	84.7	84.7
	Some difficulty	1369	7.9	8.3	93.0
	A lot of difficulty	491	2.8	3.0	96.0
	Cannot do at all	623	3.6	3.8	99.8
	Refused	27	.2	.2	99.9
	Not ascertained	2	.0	.0	100.0
	Don't know	7	.0	.0	100.0
	Total	16411	94.7	100.0	
Missing		915	5.3		
Total		17326	100.0		

MOB_5: Diff walking 1/3rd mile on level ground w/o aid or equipment

		Frequency	Percent	Valid Percent	Cumulative Percent
Valid	No difficulty	13025	75.2	82.5	82.5
	Some difficulty	1650	9.5	10.5	93.0
	A lot of difficulty	708	4.1	4.5	97.4
	Cannot do at all	347	2.0	2.2	99.6
	Refused	29	.2	.2	99.8
	Not ascertained	2	.0	.0	99.8
	Don't know	25	.1	.2	100.0
	Total	15786	91.1	100.0	
Missing		1540	8.9		
Total		17326	100.0		

Step 20. Generate a cross-tabulation of the walking distance questions: MOB_4 and MOB_5.

The syntax below produces a cross-tabulation of the two Extended Set WALKING questions: MOB_4: Difficulty walking 100 yards without equipment and MOB_5: Difficulty walking 1/3 mile without equipment to determine a single WALKING INDICATOR.

NOTE: **623** individuals who responded cannot do at all to MOB_4 were not asked MOB_5 and they do not appear in the table below. They are, however, accounted for in the WALKING indicator calculation.

CROSSTABS MOB_4 BY MOB_5.

MOB_4: Diff walking 100 yards on level ground w/o aid or equipment	MOB_5: Diff walking 1/3rd mile on level ground w/o aid or equipment							Total
	No difficulty	Some difficulty	A lot of difficulty	Cannot do at all	Refused	Not ascertained	Don't know	
No difficulty	12950	819	63	39	1	2	18	13892
Some difficulty	72	810	343	142	1	0	1	1369
A lot of difficulty	3	21	301	166	0	0	0	491
Cannot do at all (623)	0	0	0	0				
Refused	0	0	0	0	27	0	0	27
Don't know	0	0	1	0	0	0	6	7
Total	13025	1650	708	347	29	2	25	15786

Step 21. Create a WALKING INDICATOR (WALK_INDICATOR) based on the two additional walking questions MOB_4 and MOB_5.

Syntax below creates WALKING_INDICATOR based on the distribution in the cross-tabulation above.

```

COMPUTE WALK_INDICATOR = 0.
IF (MOB_4 = 1 AND (MOB_5 = 1 OR MOB_5 = 2)) WALK_INDICATOR = 1.
IF (MOB_4 = 1 AND MOB_5 = 3) OR (MOB_4 = 2 AND (MOB_5 = 1 OR MOB_5 = 2 OR MOB_5 = 3)) WALK_INDICATOR = 2.
IF (MOB_4 = 1 AND MOB_5 = 4) OR (MOB_4 = 3 AND (MOB_5 = 1 OR MOB_5 = 2 OR MOB_5 = 3)) WALK_INDICATOR = 3.
IF ((MOB_4 = 2 AND MOB_5 = 4) OR (MOB_4 = 3 AND MOB_5 = 4))
WALK_INDICATOR = 4.
    
```

Syntax below includes the 623 who responded cannot do at all to MOB_4 into the WALKING INDICATOR.

```

IF (WALK_INDICATOR = 0 AND MOB_4 = 4) WALK_INDICATOR = 4.
RECODE WALK_INDICATOR (0 = SYSMIS).
    
```

```

FREQUENCIES WALK_INDICATOR.
    
```

WALK_INDICATOR

		Frequency	Percent	Valid Percent	Cumulative Percent
Valid	1.00	13769	79.5	84.2	84.2
	2.00	1288	7.4	7.9	92.1
	3.00	364	2.1	2.2	94.3
	4.00	931	5.4	5.7	100.0
	Total	16352	94.4	100.0	
Missing		974	5.6		
Total		17326	100.0		

Step 22. Supplement Walking Indicator with information on difficulty Climbing steps (MOB_6).

Syntax below adds information from MOB_6 on difficulty climbing up or down 12 steps to create a combined Mobility Indicator (MOB_INDICATOR).

CROSSTABS WALK_INDICATOR BY MOB_6.

		MOB_6: Diff walking up or down 12 steps w/o aid or equipment						
WALK_INDICATOR		No difficulty	Some difficulty	A lot of difficulty	Cannot do at all	Refused	Don't know	Total
walk_INDICATOR2	1.00	13048	645	55	17	1	3	13769
	2.00	370	767	135	16	0	0	1288
	3.00	43	106	200	14	0	1	364
	4.00	79	242	226	384	0	0	931
Total		13540	1760	616	431	1	4	16352

According to the table above, the syntax below reclassifies:

1. 55 individuals with level **1** on the WALKING INDICATOR as level **2** on the MOBILITY INDICATOR
2. 152 individuals with level **2** on the WALKING INDICATOR as level **3** on the MOBILITY INDICATOR, and
3. 16 individuals with level **2** on the WALKING INDICATOR as level **4**. on the MOBILITY INDICATOR.

Step 23. Create a mobility indication (MOB_INDICATOR) with information garnered from cross-tabulation above.

COMPUTE MOB_INDICATOR = WALK_INDICATOR.
IF (WALK_INDICATOR = 2 AND MOB_6 = 3) MOB_INDICATOR = 3.

IF (WALK_INDICATOR = 1 AND MOB_6 = 3) MOB_INDICATOR = 2.
IF (WALK_INDICATOR = 2 AND MOB_6 = 4) MOB_INDICATOR = 4.
IF (WALK_INDICATOR = 1 AND MOB_6 = 4) MOB_INDICATOR = 3.
FREQUENCIES MOB_INDICATOR.

		MOB_INDICATOR		Valid	Cumulative
		Frequency	Percent	Percent	Percent
Valid	1.00	13697	79.1	83.8	83.8
	2.00	1192	6.9	7.3	91.1
	3.00	516	3.0	3.2	94.2
	4.00	947	5.5	5.8	100.0
	Total	16352	94.4	100.0	
Missing		974	5.6		
Total		17326	100.0		

ANXIETY

Step 24. Generate frequency distribution on ANX_1.

First, calculate frequency distributions on ANX_1: How often do you feel worried, nervous or anxious?

FREQUENCY ANX_1.

ANX_1: How often feel worried, nervous, or anxious?

		Frequency	Percent	Valid	Cumulative
				Percent	Percent
Valid	DAILY	1632	9.4	9.4	9.4
	WEEKLY	1872	10.8	10.8	20.2
	MONTHLY	1558	9.0	9.0	29.2
	A FEW TIMES A YEAR	4898	28.3	28.3	57.5
	NEVER	6714	38.8	38.8	96.2
	DON'T KNOW	652	3.8	3.8	100.0
	Total	17326	100.0	100.0	

Step 25. The syntax below recodes ANX_3 into ANX_3Y to place “SOMEWHERE BETWEEN” numerically in-between “A LITTLE” and “A LOT”.

RECODE ANX_3 (SYSMIS=SYSMIS) (0=0) (1=1) (2=3) (3=2) INTO ANX_3Y.
FREQUENCY ANX_3Y.

ANX_3Y: Level of feelings last time felt worried/nervous/anxious

		Frequency	Percent	Valid Percent	Cumulative Percent
Valid	NOT ASKED	6638	38.3	39.6	39.6
	A LITTLE	5744	33.2	34.2	73.8
	IN BETWEEN	3092	17.8	18.4	92.2
	A LOT	1189	6.9	7.1	99.3
	DON'T KNOW	113	.7	.7	100.0
	Total	16776	96.8	100.0	
Missing		550	3.2		
Total		17326	100.0		

Step 26. Generate a cross-tabulation of the anxiety Extended Set questions: ANX_1 and ANX_3Y.

The syntax below produces a cross-tabulation of ANX_1: How often you felt worried, nervous or anxious (a measure of frequency) and ANX_3Y: The level of those feeling the last time you felt worried, nervous or anxious (a measure of intensity) – used to determine a single ANXIETY INDICATOR (ANX_INDICATOR).

CROSSTABS ANX_3Y BY ANX_1.

ANX_3Y: Level of feelings last time felt worried, nervous or anxious	ANX_1: How often feel worried, nervous, or anxious?						Total
	DAILY	WEEKLY	MONTHLY	A FEW TIMES A YEAR	NEVER	DON'T KNOW	
NOT ASKED	0	0	0	0	6638	0	6638
A LITTLE	489	887	897	3417	44	10	5744
IN BETWEEN	589	725	535	1221	16	6	3092
A LOT	548	256	123	248	13	1	1189
DON'T KNOW	6	4	3	12	3	85	113
Total	1632	1872	1558	4898	6714	102	16776

Step 27. Create an ANXIETY INDICATOR (ANX_INDICATOR) based on the two anxiety questions ANX_1 and ANX_3Y.

Syntax below creates ANX_INDICATOR based on the distribution in the cross-tabulation above.

```

COMPUTE ANX_INDICATOR = 0.
IF (ANX_1 = 4 OR ANX_1 = 5) ANX_INDICATOR=1.
IF ((ANX_1 = 3) OR (ANX_1 LT 3 AND ANX_3Y=1) OR (ANX_1 = 2 AND ANX_3Y = 2))
ANX_INDICATOR = 2.
IF ((ANX_1 = 1 AND ANX_3Y = 2) OR (ANX_1 = 2 AND ANX_3Y = 3))
ANX_INDICATOR = 3.
IF (ANX_1 = 1 AND ANX_3Y = 3) ANX_INDICATOR = 4.
IF (ANX_1 = 9 OR ANX_3Y = 9) ANX_INDICATOR=9.
VALUE LABELS ANX_INDICATOR 0 'N/A' 9 "DON'T KNOW".
FREQUENCIES ANX_INDICATOR.
    
```

		ANX_INDICATOR			
		Frequency	Percent	Valid Percent	Cumulative Percent
Valid	1.00	11597	66.9	69.7	69.7
	2.00	3656	21.1	22.0	91.6
	3.00	845	4.9	5.1	96.7
	4.00	548	3.2	3.3	100.0
	Total	16646	96.1	100.0	
Missing	Don't know	680	3.9		
Total		17326	100.0		

DEPRESSION

Step 28. Generate frequency distribution on DEP_1.

First, calculate frequency distributions on DEP_1: How often do you feel depressed?

```

FREQUENCY DEP_1.
    
```

DEP_1: How often do you feel depressed?

		Frequency	Percent	Valid Percent	Cumulative Percent
Valid	DAILY	756	4.4	4.4	4.4
	WEEKLY	926	5.3	5.3	9.7
	MONTHLY	1038	6.0	6.0	15.7
	A FEW TIMES A YEAR	4012	23.2	23.2	38.9
	NEVER	9929	57.3	57.3	96.2
	DON'T KNOW	665	3.8	3.8	100.0
	Total	17326	100.0	100.0	

Step 29. The syntax below recodes DEP_3 into DEP_3Y to place “SOMEWHERE BETWEEN” numerically in-between “A LITTLE” and “A LOT”.

**RECODE DEP_3 (SYSMIS=SYSMIS) (0=0) (1=1) (2=3) (3=2) INTO DEP_3Y.
FREQUENCY DEP_3Y.**

DEP_3Y: Level of feelings last time felt depressed

		Frequency	Percent	Valid Percent	Cumulative Percent
Valid	NOT ASKED	9785	56.5	58.3	58.3
	A LITTLE	3847	22.2	22.9	81.3
	IN BETWEEN	2051	11.8	12.2	93.5
	A LOT	970	5.6	5.8	99.3
	DON'T KNOW	118	.7	.7	100.0
	Total	16771	96.8	100.0	
Missing		555	3.2		
Total		17326	100.0		

Step 30. Generate a cross-tabulation of the depression Extended Set questions: DEP_1 and DEP_3Y.

The syntax below produces a cross-tabulation of DEP_1: How often do you feel depressed (a measure of frequency) and DEP_3Y: The level of those feeling the last time you felt depressed (a measure of intensity) – used to determine a single DEPRESSION INDICATOR (DEP_INDICATOR).

CROSSTABS DEP_3Y BY DEP_1.

DEP_3Y: Level of feelings last time felt depressed		DEP_1: How often do you feel depressed?					DONT KNOW	Total
		DAILY	WEEKLY	MONTHLY	A FEW TIMES A YEAR	NEVER		
Level of feelings last time felt depressed	NOT ASKED	0	0	0	0	9785	0	9785
	A LITTLE	161	346	548	2708	72	12	3847
	IN BETWEEN	209	384	378	1042	35	3	2051
	A LOT	381	191	112	248	35	3	970
	DONT KNOW	5	5	0	14	2	92	118
Total		756	926	1038	4012	9929	110	16771

Step 31. Create a DEPRESSION INDICATOR (DEP_INDICATOR) based on the two depression questions DEP_1 and DEP_3Y.

Syntax below creates DEP_INDICATOR based on the distribution in the cross-tabulation above.

```

COMPUTE DEP_INDICATOR = 0.
IF (DEP_1 = 4 OR DEP_1 = 5) DEP_INDICATOR=1.
IF ((DEP_1 = 3) OR (DEP_1 LT 3 AND DEP_3Y=1) OR (DEP_1 = 2 AND DEP_3Y = 2))
DEP_INDICATOR = 2.
IF ((DEP_1 = 1 AND DEP_3Y = 2) OR (DEP_1 = 2 AND DEP_3Y = 3)) DEP_INDICATOR =
3.
IF (DEP_1 = 1 AND DEP_3Y = 3) DEP_INDICATOR = 4.
IF (DEP_1 = 9 OR DEP_3Y = 9) DEP_INDICATOR = 9.
VALUE LABELS DEP_INDICATOR 0 'N/A' 9 "DON'T KNOW".

FREQUENCIES DEP_INDICATOR.
    
```

DEP_INDICATOR

		Frequency	Percent	Valid Percent	Cumulative Percent
Valid	1.00	13925	80.4	83.7	83.7
	2.00	1929	11.1	11.6	95.3
	3.00	400	2.3	2.4	97.7
	4.00	381	2.2	2.3	100.0
	Total	16635	96.0	100.0	
Missing	Don't know	691	4.0		
Total		17326	100.0		

PAIN

Step 32. Generate frequency distribution on PAIN_2.

First, calculate frequency distributions on PAIN_2: Frequency of pain in the past 3 months.

FREQUENCY PAIN_2.

PAIN_2: Frequency of pain in past 3 months

		Frequency	Percent	Valid Percent	Cumulative Percent
Valid	Never	6636	38.3	38.3	38.3
	Some days	6556	37.8	37.8	76.1
	Most days	1227	7.1	7.1	83.2
	Every day	2245	13.0	13.0	96.2
	Refused	63	.4	.4	96.5
	Not ascertained	564	3.3	3.3	99.8
	Don't know	35	.2	.2	100.0
	Total	17326	100.0	100.0	

Step 33. The syntax below recodes PAIN_3 into PAIN_3Y to place “SOMEWHERE BETWEEN” numerically in-between “A LITTLE” and “A LOT”.

RECODE PAIN_4 (SYSMIS=SYSMIS) (0=0) (1=1) (2=3) (3=2) INTO PAIN_4Y.
IF (PAIN_2 = 1) PAIN_4Y=0.

FREQUENCIES PAIN_4Y.

PAIN_4Y: How much pain you had last time you had pain?

		Frequency	Percent	Valid Percent	Cumulative Percent
Valid	NOT ASKED	6636	38.3	39.6	39.6
	A LITTLE	4865	28.1	29.0	68.6
	IN BETWEEN	3296	19.0	19.7	88.3
	A LOT	1869	10.8	11.2	99.4
	DON'T KNOW	96	.6	.6	100.0
	Total	16762	96.7	100.0	
Missing		564	3.3		
Total		17326	100.0		

Step 34. Generate a cross-tabulation of the PAIN Extended Set questions: PAIN_2 and PAIN_4Y.

The syntax below produces a cross-tabulation of PAIN_2: Frequency of pain in the past 3 months and PAIN_4Y: How much pain you has the last time you had pain (a measure of intensity) – used to determine a single PAIN INDICATOR (P_INDICATOR).

CROSSTABS PAIN_4Y BY PAIN_2.

PAIN_4Y: How much pain you had last time you had pain		PAIN_2: Frequency of pain in past 3 months					Total
		Never	Some days	Most days	Every day	Don't know	
pain_4y	NOT ASKED	6636	0	0	0	0	6636
	A LITTLE	0	4136	323	401	5	4865
	IN BETWEEN	0	1772	624	896	4	3296
	A LOT	0	645	278	944	2	1869
	DON'T KNOW	0	3	2	4	87	96
Total		6636	6556	1227	2245	98	16762

Step 35. Create a PAIN INDICATOR (P_INDICATOR) based on the two PAIN questions PAIN_2 and PAIN_4Y.

Syntax below creates P_INDICATOR based on the distribution in the cross-tabulation above.

COMPUTE P_INDICATOR = 0.

IF (PAIN_2 = 1) OR (PAIN_4Y = 1 AND (PAIN_2 = 2 OR PAIN_2 = 3)) P_INDICATOR = 1.

IF ((PAIN_2 = 2 AND (PAIN_4Y = 2 OR PAIN_4Y = 3)) OR (PAIN_2 = 3 AND PAIN_4Y = 2) OR (PAIN_2 = 4 AND PAIN_4Y = 1)) P_INDICATOR = 2.

IF (PAIN_2 = 3 AND PAIN_4Y = 3) OR (PAIN_2 = 4 AND PAIN_4Y = 2) P_INDICATOR = 3.

IF (PAIN_2 = 4 AND PAIN_4Y = 3) P_INDICATOR = 4.

RECODE P_INDICATOR (0=SYSMIS).

FREQUENCIES P_INDICATOR.

		P_INDICATOR			
		Frequency	Percent	Valid Percent	Cumulative Percent
Valid	1.00	11095	64.0	66.6	66.6
	2.00	3442	19.9	20.7	87.3
	3.00	1174	6.8	7.0	94.3
	4.00	944	5.4	5.7	100.0
	Total	16655	96.1	100.0	
Missing		671	3.9		
Total		17326	100.0		

FATIGUE (Tired)

Step 36. Generate frequency distribution on FATIGUE Extended Set questions Tired_1, Tired_2 and Tired_3.

First, calculate frequency distributions on TIRED_1: How often you felt tired in the past 3 months.

Also recode TIRED_1, TIRED_2 AND TIRED_3 where refused (7) non ascertained (8) and Don't know (9) are combined into (9)

RECODE TIRED_1 (7,8,9=9).

RECODE TIRED_2 (7,8,9=9).

RECODE TIRED_3 (7,8,9=9).

FREQUENCIES TIRED_1.

TIRED_1: How often felt very tired or exhausted in past 3 months

		Frequency	Percent	Valid Percent	Cumulative Percent
Valid	Never	5619	32.4	32.4	32.4
	Some days	8391	48.4	48.4	80.9
	Most days	1632	9.4	9.4	90.3
	Every day	1019	5.9	5.9	96.2
	Don't know	665	3.8	3.8	100.0
	Total	17326	100.0	100.0	

Step 37. Recode Tired_2 to 0 (not asked) if Tired_1 is 1 (Never).

If response to TIRED_1 is 1: Never, then TIRED_2 is not asked. This is recoded so these individuals are included in the syntax below.

```
IF (TIRED_1 = 1) TIRED_2=0.
VALUE LABELS TIRED_2 0 'NOT ASKED'.
FREQUENCIES TIRED_2.
```

TIRED_2: How long most recent tired or exhausted feelings lasted?

		Frequency	Percent	Valid Percent	Cumulative Percent
Valid	Not asked	5619	32.4	33.5	33.5
	Some of the day	8036	46.4	48.0	81.5
	Most of the day	1955	11.3	11.7	93.2
	All of the day	1036	6.0	6.2	99.3
	Don't know	111	.6	.7	100.0
	Total	16757	96.7	100.0	
Missing		569	3.3		
Total		17326	100.0		

Step 38. The syntax below recodes TIRED_3 into TIRED_3Y to place “SOMEWHERE BETWEEN” numerically in-between “A LITTLE” and “A LOT”.

```
RECODE TIRED_3 (SYSMIS=SYSMIS) (0=0) (1=1) (2=3) (3=2) (9=9) INTO TIRED_3Y.
```

Also, if response to TIRED_1 is 1: Never, then TIRED_3 is not asked. This is recoded so these individuals are included in the syntax below.

```
IF (TIRED_1 = 1) TIRED_3Y=0.
VALUE LABELS TIRED_3Y 0 'NOT ASKED' 1 'A LITTLE' 2 'IN BETWEEN' 3 'A LOT' 9
'DON'T KNOW'.
VARIABLE LABELS TIRED_3Y 'LEVEL OF TIREDNESS'.
FREQUENCIES TIRED_3Y.
```

TIRED_3Y: Level of tiredness

		Frequency	Percent	Valid Percent	Cumulative Percent
Valid	NOT ASKED	5619	32.4	33.5	33.5
	A LITTLE	4912	28.4	29.3	62.8
	IN BETWEEN	4030	23.3	24.0	86.9
	A LOT	2087	12.0	12.5	99.3
	DON'T KNOW	109	.6	.7	100.0
	Total	16757	96.7	100.0	
Missing		569	3.3		
Total		17326	100.0		

Step 39. Generate a cross-tabulation of the FATIGUE Extended Set questions: TIRED_1, TIRED_2 and TIRED_3Y.

The syntax below produces a cross-tabulation of TIRED_1: How often you felt tired or exhausted in the past 3 months (a measure of frequency) and TIRED_2: How long those feelings lasted (a measure of duration) and TIRED_3Y: The level of tiredness (a measure of intensity) – used to determine a single TIRED INDICATOR (T_INDICATOR).

CROSSTABS TIRED_2 BY TIRED_1 BY TIRED_3Y.

TIRED_3Y: Level of tiredness: Intensity	TIRED_2: How long feelings lasted: Duration	TIRED_1: How often felt very tired or exhausted in past 3 months: Frequency				Total
		Never	Some days	Most days	Every day	
Not asked	Not asked	5619				5619
A little	Some of the day		4066	264	124	4454
	Most of the day		252	73	27	352
	All of the day		68	15	18	101
In between	Some of the day		2224	400	166	2791
	Most of the day		497	266	123	887
	All of the day		194	71	84	349
A lot	Some of the day		536	165	84	785
	Most of the day		297	255	160	713
	All of the day		237	118	230	585
TOTAL		5619	8371	1627	1016	16633

Step 40. Create a **FATIGUE INDICATOR (T_INDICATOR)** based on the three **FATIGUE** questions **TIRED_1**, **TIRED_2** and **TIRED_3Y**.

Syntax below creates **T_INDICATOR** based on the distribution in the cross-tabulation above.

```

COMPUTE T_INDICATOR = 0.
IF (TIRED_1 = 1) T_INDICATOR=1.
IF (TIRED_1 = 2 AND TIRED_2 = 1 AND TIRED_3Y = 1) T_INDICATOR = 1.
IF (TIRED_1 = 3 AND TIRED_2 = 1 AND TIRED_3Y = 1) T_INDICATOR = 1.

IF (T_INDICATOR NE 1 AND TIRED_3Y LT 3) T_INDICATOR = 2.
IF (T_INDICATOR NE 1 AND (TIRED_3Y = 3 AND TIRED_2 = 1)) T_INDICATOR = 2.
IF (T_INDICATOR NE 1 AND (TIRED_3Y = 3 AND TIRED_2 = 2 AND TIRED_1 = 2))
T_INDICATOR = 2.

IF ((T_INDICATOR NE 1 AND T_INDICATOR3 NE 2) AND TIRED_1 = 2) T_INDICATOR
= 3.
IF (TIRED_1 GE 3 AND TIRED_2 = 2 AND TIRED_3Y = 3) OR (TIRED_1 GE 3 AND
TIRED_2 = 3 AND TIRED_3Y = 3) OR (TIRED_1 GE 3 AND TIRED_2 = 1 AND TIRED_3Y
= 5) OR (TIRED_1 = 3 AND TIRED_2 = 2 AND TIRED_3Y = 5) T_INDICATOR = 3.
IF (TIRED_1 GE 3 AND TIRED_2 = 2 AND TIRED_3Y = 3) T_INDICATOR = 3.

IF (T_INDICATOR NE 1 AND T_INDICATOR3 NE 2 AND T_INDICATOR3 NE 3 AND
(TIRED_1 = 3 OR TIRED_1 = 4)) T_INDICATOR = 4.

IF (TIRED_1 GE 7 OR TIRED_2 GE 7 OR TIRED_3Y =9) T_INDICATOR = 9.

VALUE LABELS T_INDICATOR 0 'N/A' 9 "DON'T KNOW".
FREQUENCIES T_INDICATOR.
    
```

		T_indicator		Valid Percent	Cumulative Percent
		Frequency	Percent		
Valid	1.00	9949	57.4	59.8	59.8
	2.00	3864	22.3	23.2	83.0
	3.00	2312	13.3	13.9	96.9
	4.00	508	2.9	3.1	100.0
	Total	16633	96.0	100.0	
Missing	Don't know	693	4.0		
Total		17326	100.0		

Creating Disability Status Indicators

Type of Disability Indicator		Number of Questions
SS_1	Short Set (SS)	6
Extended Set		
ES_1	SS + <u>Hearing-indicator</u>, <u>Mobility-indicator</u>, <u>Cognition-indicator</u>, <u>Upper Body-indicator</u> + PFAD (4)*	25
ES_2	SS + <u>Hearing-indicator</u>, <u>Mobility-indicator</u>, <u>Cognition-indicator</u>, <u>Upper Body-indicator</u> + AD (4)†	20
Short Set Enhanced		
ES_3	SS + Upper Body-indicator + AD (4)†	12

* PFAD (4): Pain, Fatigue, Anxiety and Depression Indicators at level 4

† AD (4): Anxiety and Depression Indicators at level 4

SS_1: WG Short Set Disability Indicator based on the 6 short set questions.

The syntax below calculates the WG Short Set Disability Indicator based on the six short set questions **SS_1** at the recommended cut-off. The level of inclusion is: at least 1 domain/question is coded A LOT OF DIFFICULTY or CANNOT DO AT ALL.

FREQUENCIES VIS_SS HEAR_SS MOB_SS COM_SS UB_SS COG_SS.

COMPUTE SS_1 = 0.

IF (VIS_SS GE 7 AND HEAR_SS GE 7 AND MOB_SS GE 7 AND COM_SS GE 7 AND UB_SS GE 7 AND COG_SS GE 7) SS_1 = 9.

IF ((VIS_SS = 3 OR VIS_SS = 4) OR (HEAR_SS= 3 OR HEAR_SS = 4) OR (MOB_SS= 3 OR MOB_SS = 4) OR (COM_SS= 3 OR COM_SS = 4) OR (UB_SS = 3 OR UB_SS = 4) OR (COG_SS = 3 OR COG_SS = 4)) SS_1 = 1.

FREQUENCIES SS_1.

RECODE SS_1 (9=SYSMIS).

FREQUENCIES SS_1.

SS_1: WG Short Set Disability Identifier

		Weighted Estimate	Unweighted Count
% of Total	WITHOUT DISABILITY	90.8%	15454
	WITH DISABILITY	9.2%	1872
	Total	100.0%	17326

ES_1: SS_1 + Hearing-indicator, Mobility-indicator, Cognition-indicator + Upper Body-indicator + PFAD (4).

The syntax below calculates an Extended Set Disability Indicator (**ES_1**) based on 25 questions at the recommended cut-off. The level of inclusion is: at least 1 domain/question is coded A LOT OF DIFFICULTY or CANNOT DO AT ALL for the six short set question; severity levels 3 or 4 for the Hearing-, Mobility-, Cognition- and Upper body-Indicators; and severity level 4 for Pain-, Fatigue-, Anxiety- and Depression-Indicators.

```

COMPUTE ES_1 = 0.
IF (SS_1 GE 7 AND (H_INDICATOR LT 1 OR H_INDICATOR GT 4) AND
(MOB_INDICATOR LT 1 OR MOB_INDICATOR GT 4) AND COM_SS GE 7 AND UB_SS
GE 7 AND (COG_INDICATOR LT 1 OR COG_INDICATOR GT 4) AND (UB_INDICATOR
LT 1 OR UB_INDICATOR GT 4) AND (SYSMIS(P_INDICATOR)) AND (T_INDICATOR
LT 1 OR T_INDICATOR GT 4) AND (ANX_INDICATOR LT 1 OR ANX_INDICATOR GT
4) AND (DEP_INDICATOR LT 1 OR DEP_INDICATOR GT 4)) ES_1 = 9.

IF (SS_1 = 1 OR (H_INDICATOR = 3 OR H_INDICATOR = 4) OR (MOB_INDICATOR = 3
OR MOB_INDICATOR = 4) OR
(COG_INDICATOR = 3 OR COG_INDICATOR = 4) OR (UB_INDICATOR = 3 OR
UB_INDICATOR = 4) OR P_INDICATOR = 4 OR T_INDICATOR = 4 OR
ANX_INDICATOR = 4 OR DEP_INDICATOR = 4) ES_1 = 1.

FREQUENCIES ES_1.
RECODE ES_1 (9=SYSMIS).
FREQUENCIES ES_1.
    
```

ES_1: WG-ES Disability Indicator based on 11 domains and 25 questions

		Weighted Estimate	Unweighted Count
% of Total	WITHOUT DISABILITY	82.3%	13823
	WITH DISABILITY	17.7%	3503
	Total	100.0%	17326

ES_2: SS_1 + Hearing-indicator, Mobility-indicator, Cognition-indicator + Upper Body-indicator + AD (4)

The syntax below calculates an Extended Set Disability Indicator (**ES_2**) based on 20 questions at the recommended cut-off. The level of inclusion is: at least 1 domain/question is coded A LOT OF DIFFICULTY or CANNOT DO AT ALL for the six short set question; severity levels 3 or 4 for the Hearing-, Mobility-, Cognition- and Upper body-Indicators; and severity level 4 for Anxiety- and Depression-Indicators.

```

COMPUTE ES_2 = 0.
IF (SS_1 GE 7 AND (H_INDICATOR LT 1 OR H_INDICATOR GT 4) AND
(MOB_INDICATOR LT 1 OR MOB_INDICATOR GT 4) AND COM_SS GE 7 AND UB_SS
GE 7 AND (COG_INDICATOR LT 1 OR COG_INDICATOR GT 4) AND (UB_INDICATOR
LT 1 OR UB_INDICATOR GT 4) AND (ANX_INDICATOR LT 1 OR ANX_INDICATOR GT
4) AND (DEP_INDICATOR LT 1 OR DEP_INDICATOR GT 4)) ES_2 = 9.

IF (SS_1 = 1 OR (H_INDICATOR = 3 OR H_INDICATOR = 4) OR (MOB_INDICATOR = 3
OR MOB_INDICATOR = 4) OR
(COG_INDICATOR = 3 OR COG_INDICATOR = 4) OR (UB_INDICATOR = 3 OR
UB_INDICATOR = 4) OR ANX_INDICATOR = 4 OR DEP_INDICATOR = 4) ES_2 = 1.

FREQUENCIES ES_2.
RECODE ES_2 (9=SYSMIS).
FREQUENCIES ES_2.
    
```

ES_2: WG-ES Disability Indicator based on 9 domains and 20 questions

		Weighted Estimate	Unweighted Count
% of Total	WITHOUT DISABILITY	84.6%	14222
	WITH DISABILITY	15.4%	3104
	Total	100.0%	17326

ES_3: SS_1 + Upper Body-indicator + AD (4)

The syntax below calculates the WG Short Set ENHANCED Disability Indicator (**ES_3**) based on the 12 questions at the recommended cut-off. The level of inclusion is: at least 1 domain/question is coded A LOT OF DIFFICULTY or CANNOT DO AT ALL for the six short set question; severity levels 3 or 4 for the Upper body-Indicators; and severity level 4 for Anxiety- and Depression-Indicators.

```

COMPUTE ES_3 = 0.
IF (VIS_SS GE 7 AND HEAR_SS GE 7 AND MOB_SS GE 7 AND COM_SS GE 7 AND
UB_SS GE 7 AND COG_SS GE 7 AND (ANX_INDICATOR LT 1 OR ANX_INDICATOR
GT 4) AND (DEP_INDICATOR LT 1 OR DEP_INDICATOR GT 4)) ES_3 = 9.
    
```

IF ((VIS_SS = 3 OR VIS_SS = 4) OR (HEAR_SS= 3 OR HEAR_SS = 4) OR (MOB_SS= 3 OR MOB_SS = 4) OR (COM_SS= 3 OR COM_SS = 4) OR (UB_SS = 3 OR UB_SS = 4) OR (COG_SS = 3 OR COG_SS = 4) OR ANX_INDICATOR = 4 OR DEP_INDICATOR = 4) ES_3 = 1.

FREQUENCIES ES_3.

RECODE ES_3 (9=SYSMIS).

FREQUENCIES ES_3.

ES_3: WG-SS Enhanced Disability Indicator based on 9 domains and 12 questions

		Weighted Estimate	Unweighted Count
% of Total	WITHOUT DISABILITY	88.1%	14942
	WITH DISABILITY	11.9%	2384
	Total	100.0%	17326

Postscript: Why exclude Pain and fatigue?

Of note is the exclusion of the pain and fatigue domains from several of the Disability Identifiers above. There has been considerable discussion within the WG on these domains. They are not, strictly speaking, domains of functioning – and our analyses indicated that they are both highly correlated with other domains – and that the rates of disability with the inclusion of these domains can be very high. Finally, in terms of international comparability, these domains are less universal; that is, they are more susceptible to local, socio-cultural influences than other domains of functioning.

For these reasons, we chose to exclude them from several of these analyses, though they can be included in supplemental analyses carried out by NSOs on a national basis.